A. ACSL SIMULATION MODEL PROGRAM

```
! Model Program

! CASE : CASCADE TEMPERATURE - BASE
! TUNING : DPC BASED

PROGRAM


CINTERVAL    CINT  = 5      ! Communication interval [s]
MAXTERVAL    MAXT  = 0.05   ! Upper bound of integration step [s]
ALGORITHM    IALG  = 5      ! Integration algorithm
                            ! Runge-Kutta fourth order
NSTEPS       NSTEP = 1      ! Number of integration steps per  communication interval


! ==========================================================================
! DATA
! ==========================================================================

! Refrigerant inventory [kg]

CONSTANT W  = 6500

! Vessels sizes [m3]

CONSTANT V1 = 20                ! LP evaporator
CONSTANT V2 = 3.393             ! IP evaporator
```

```
CONSTANT V3 = 18.85                 ! Receiver
CONSTANT VC = 4.15                  ! Condenser

! Coefficients for curve fittings for first compressor performance

CONSTANT C11 = 1.9928
CONSTANT C12 = 16791
CONSTANT C13 = 8756.4

! Coefficients for curve fittings for second compressor performance

CONSTANT C21 = 0.45615
CONSTANT C22 = 10296
CONSTANT C23 = 2956.2
CONSTANT C24 = 0.816051
CONSTANT C25 = 0.27585

! Coefficients for curve fittings for first compressor isentropic efficiency

CONSTANT E11 = 6.47E-5
CONSTANT E12 = 0.353
CONSTANT E13 = 7.101E-4
CONSTANT E14 = 6.5963

! Coefficients for curve fittings for second compressor isentropic efficiency

CONSTANT E21 = 4.098E-5
CONSTANT E22 = 0.364
CONSTANT E23 = 1.121E-3
CONSTANT E24 = 12.445

! Combined overall heat transfer coefficients and heat transfer areas
! in the two evaporators and the condenser [J/(s.K)]

CONSTANT U1A1 = 146.066             ! LP evaporator
CONSTANT U2A2 = 5.5479              ! IP evaporator
CONSTANT U3A3 = 420.643             ! Condenser

! Valve constants [kg/(s.bar0.5)]
```

```
CONSTANT CV1 = 3.39814          ! Vapour valve
CONSTANT CV2 = 2.233338         ! First liquid valve
CONSTANT CV3 = 1.85435          ! Second liquid valve

! Combined process stream flowrate and specific heat capacity [J/(s.K)]

CONSTANT FCP1 = 111.394         ! LP evaporator
CONSTANT FCP2 = 24.32           ! IP evaporator

! Process stream inlet temperatures [K]
! Note that TP1I is not included in a CONSTANT statement to avoid
! a warning message when it is step-change

TP1I = 235.2                    ! Process stream in LP evaporator
CONSTANT TP2I = 280.4           ! Process stream in IP evaporator
CONSTANT TP3I = 303.0           ! Cooling air in the condenser

! Antoine Equation coefficients for propylene

CONSTANT A = 9.0825
CONSTANT B = 1807.53
CONSTANT C = 26.15

! General constants

CONSTANT R  = 8.314             ! Gas constant [J/(mol.K)]
CONSTANT MW = 42.081            ! Propylene molecular weight [kg/kmol]
CONSTANT G  = 9.81              ! Gravity acceleration [m/s2]

! Specific heat capacity equation coefficients for propylene

CONSTANT C1 = 3.707
CONSTANT C2 = 0.01
CONSTANT C3 = 23.439
CONSTANT C4 = 0.001
CONSTANT C5 = -11.594
CONSTANT C6 = 2.2033E-3


! ==================================================================
```

```
! INITIAL CONDITIONS
! ======================================================================

! Initial values are needed for all INTEG and IMPLC statements in the
! model

! Initial values for state variables

CONSTANT W1ic = 2000            ! Refrigerant holdup in LP evaporator [kg]
CONSTANT W2ic = 400             ! Refrigerant holdup in IP evaporator [kg]
CONSTANT T3ic = 312.325         ! Liquid refrigerant temperature in the receiver [K]
CONSTANT H1ic = 1.20644E6       ! Heat content in LP evaporator [J]
CONSTANT H2ic = 2.89417E5       ! Heat content in IP evaporator [J]
CONSTANT W4ic = 540.333         ! Vapour refrigerant holdup in the condenser [kg]

! Initial values for variables evaluated using implicit ACSL solver IMPLC

CONSTANT T1ic = 222.88          ! LP evaporator temperature [K]
CONSTANT T2ic = 262.508         ! IP evaporator temperature [K]
CONSTANT PAic = 3.74756         ! Inter-stage mixing node pressure [bar]
CONSTANT TAic = 292.889         ! Discharge temperature from first compressor [K]
CONSTANT TBic = 281.941         ! Inter-stage mixing node temperature [K]
CONSTANT TCic = 360.138         ! Discharge temperature from second compressor [K]
CONSTANT TDic = 312.325         ! Refrigerant temperature in the condenser [K]
CONSTANT HH1ic = 7363           ! Head in first compressor [m]
CONSTANT HH2ic = 9298           ! Head in second compressor [m]


! ======================================================================
! CONTROLLERS SETTINGS
! ======================================================================

! Controllers gain factors

CONSTANT K1 = 2.86              ! L1 controller
CONSTANT K2 = 10.0              ! L2 controller
CONSTANT K3 = -1.0              ! P1 slave controller
CONSTANT K4 = -3.33             ! P2 controller
CONSTANT K5 = -386.67           ! P3 controller
```

```
! Controllers integral times [s]

CONSTANT TI1 = 37.61            ! L1 controller
CONSTANT TI2 = 37.61            ! L2 controller
CONSTANT TI3 = 37.61            ! P1 slave controller
CONSTANT TI4 = 37.61            ! P2 controller
CONSTANT TI5 = 37.61            ! P3 controller

! Cascade temperature controller

CONSTANT KT=0.2                 ! gain factor
CONSTANT TIT=37.61              ! Integral time [s]

! Initial values for errors, needed for INTEG statements

CONSTANT IER1ic = 0
CONSTANT IER2ic = 0
CONSTANT IER3ic = 0
CONSTANT IER4ic = 0
CONSTANT IER5ic = 0
CONSTANT IERTP1ic = 0
CONSTANT IERTP2ic = 0

! Initial values for the manipulated variables

CONSTANT XV2ic  = 0.7           ! L1-controller valve opening
CONSTANT XV3ic  = 0.7           ! L2-controller valve opening
CONSTANT Nic    = 0.979845      ! Compressor's speed
CONSTANT XV1ic  = 0.7           ! P2-controller valve opening
CONSTANT FCP3ic = 232.3256      ! Combined cooling air flowrate
                                ! and specific heat capacity [J/(s.K)]

! Set points

CONSTANT L1SP=3.304             ! L1 [m3]
CONSTANT L2SP=0.6932            ! L2 [m3]
CONSTANT P2SP=4.2               ! P2 [bar]
CONSTANT P3SP=15.9              ! P3 [bar]
CONSTANT L3SP=7.61845           ! L3 [m3]
CONSTANT TP1OSP = 226.2         ! TP1O [K]
```

```
CONSTANT TP2OSP = 276.751          ! TP2O [K]


! ==========================================================================
! ==========================================================================

INITIAL

! This section includes the evaluation of the initial values for
! the conditions in the LP and IP evaporators

! LP evaporator

P1ic = EXP (A - B/(T1ic -C))
Z1ic = 0.955 + 0.0*T1ic
VF1ic = 8.05418E-4 + 3.934E-6*T1ic
WV1ic = (P1ic*MW*100/(Z1ic*R*T1ic))*((V1-VF1ic*W1ic)/(1-(VF1ic*P1ic*MW/ &
     (Z1ic*R*T1ic))))

! IP evaporator

P2ic = EXP (A - B/(T2ic -C))
Z2ic = 0.566 + 1.26E-3*TBic
VF2ic = 5.91479E-4 + 4.777E-6*T2ic
WV2ic = (P2ic*MW*100/(Z2ic*R*T2ic))*((V2-VF2ic*W2ic)/(1-(VF2ic*P2ic*MW/ &
     (Z2ic*R*T2ic))))

END             ! OF INITIAL


! ==========================================================================
! ==========================================================================

DYNAMIC


! ==========================================================================
! ==========================================================================

! This part contains the inclusion of a step change in the
```

```
! process stream input temperature TP1I at time SS

! Time of step change occurrence

SS=1500

! Name of DISCRETE section on which a step change is performed

SCHEDULE STEADY .AT.SS

DISCRETE STEADY

TP1I=237.2

END                ! OF DISCRETE

! Note that to include more than one step change, a DISCRETE section
! is required for each

! =========================================================================
! =========================================================================


DERIVATIVE

! State variables

W1 = INTEG(DW1,W1ic)          ! Refrigerant holdup in LP evaporator [kg]
W2 = INTEG(DW2,W2ic)          ! Refrigerant holdup in IP evaporator [kg]
T3 = INTEG(DT3,T3ic)          ! Liquid refrigerant temperature in the receiver [K]
H1 = INTEG(DH1,H1ic)          ! Heat content in LP evaporator [J]
H2 = INTEG(DH2,H2ic)          ! Heat content in IP evaporator [J]
W4 = INTEG(DW4,W4ic)          ! Vapour refrigerant holdup in the condenser [kg]

! Derivatives

DW1 = FL2 - FG1
DW2 = FL3 - FL2 - FG2
DT3 = FL4 * (TD - T3) /W3
DH1 = HFL2*FL2 - HFG1*FG1 + Q1
```

```
DH2 = HFL3*FL3 - HFL2*FL2 - HFG2*FG2 + Q2
DW4 = FG3 - FL4

! ------------------------------------------------------------------------------------------------

! Liquid refrigerant mass flowrates [kg/s]

FL2 = XV2*CV2*SQRT(P2-P1)
FL3 = XV3*CV3*SQRT(P3-P2)
FL4 = -QC/(HFG3-HFL4)

! Vapour refrigerant mass flow rates [kg/s]

FG1 = FG1S * MW*P1*100/(T1*R*Z1)
FG2 = XV1*CV1*SQRT(P2-PA)
FG3 = FG3S * MW*PA*100/(TB*R*Z2)

! Vapour refrigerant volumetric flow rate [m3/s]
! Relationships obtained by curve fitting

FG1S = (N**1.56) * (C11*HS1 - C12)/(HS1 - C13)
TX  = C21/(TAN((C22-HS2)/C23))
FG3S = (N**1.79) * (C24 + C25 * LOG10(SQRT(TX**2+1)-TX))

! ------------------------------------------------------------------------------------------------

! Inter-stage mixing node temperature [K]

TB = (TA*FG1 + T2*FG2)/(FG1+FG2)

! ------------------------------------------------------------------------------------------------

! Pressures in the vessels [bar]

! Plin is defined as it is needed to be used in the linearisation of the model

CONSTANT Plin=0
P1 = EXP(A - B/(T1-C)) + Plin

P2 = EXP(A - B/(T2-C))
```

```
P3 = EXP(A - B/(TD-C))

! ------------------------------------------------------------------------------------------------------------------------

! Q : Heat loads [J/s]
! TPiO : Process stream outlet temperature [K]

! LP evaporator

Q1 = FCP1 * (TP1I - TP1O)
TP1O = (1-A1) * T1 + A1 * TP1I
A1 = EXP (-U1A1/FCP1)

! IP evaporator

Q2 = FCP2 * (TP2I - TP2O)
TP2O = (1-A2) * T2 + A2 * TP2I
A2 = EXP (-U2A2/FCP2)

! Condenser

QC = FCP3 * (TP3I - TP3O)
TP3O = (1-A3) * TD + A3 * TP3I
A3 = EXP (-U3A3/FCP3)

! ------------------------------------------------------------------------------------------------------------------------

! First compressor calculations

! Scaled head [m]

HS1 = HH1/N**2.19

! Polytropic efficiency
! Relationship obtained by curve fitting

ETA1 = E11*HS1 + E12 -10**(E13*HS1-E14)

! Specific heat capacity [J/(kg.K)]
```

```
CPI1 = C1 + C2*T1 * (C3 + C4*T1 * (C5 + C6*T1))
CPO1 = C1 + C2*TA * (C3 + C4*TA * (C5 + C6*TA))

! Evaluating k constant used in calculating the discharge temperature

G1 = 0.5 * (CPI1/(CPI1-R) + CPO1/(CPO1-R))
POL1 = ETA1 * G1 /(G1 - 1)

! Discharge temperature [K], using implicit solver

RESIDTA = TA - T1*((PA/P1)**(1/POL1))
TA = IMPLC(RESIDTA,TAic)

! Compressor's head [m], using implicit solver

RESIDHH1 = HH1 - POL1 * (R*1000/(G*MW)) * (TA - T1)
HH1 = IMPLC(RESIDHH1,HH1ic)

! Second compressor calculations

! Scaled head [m]

HS2 = HH2/N**2.11

! Polytropic efficiency
! Relationship obtained by curve fitting

ETA2 = E21*HS2 + E22 - 10**(E23*HS2-E24)

! Specific heat capacity [J/(kg.K)]

CPI2 = C1 + C2*TB * (C3 + C4*TB * (C5 + C6*TB))
CPO2 = C1 + C2*TC * (C3 + C4*TC * (C5 + C6*TC))

! Evaluating k constant used in calculating the discharge temperature

G2 = 0.5 * (CPI2/(CPI2-R) + CPO2/(CPO2-R))
POL2 = ETA2 * G2 /(G2 - 1)

! Discharge temperature [K], using implicit solver
```

```
RESIDTC = TC - TB*((P3/PA)**(1/POL2))
TC = IMPLC(RESIDTC,TCic)

! Compressor's head [m], using implicit solver

RESIDHH2 = HH2 - POL2 * (R*1000/(G*MW)) * (TC - TB)
HH2 = IMPLC(RESIDHH2,HH2ic)

! --------------------------------------------------------------------------------------------------------------------

! Vapour refrigerant streams specific enthalpies [J/kg]

HFG1 = 747.441 + 1.36908*T1
HFG2 = 663.375 + 1.63599*T2
HFG3 = 474.018 + 2.09759*TC

! Liquid refrigerant streams specific enthalpies [J/kg]

HFL2 = 7.16036 + 2.63768*T2
HFL3 = 95.5601 + 2.32367*T3
HFL4 = 95.5601 + 2.32367*TD

! Refrigerant specific enthalpy in LP evaporator [J/kg]

HW1V = 747.441 + 1.36908*T1          ! Liquid
HW1L = 7.16036 + 2.63768*T1          ! Vapour

! Refrigerant specific enthalpy in IP evaporator [J/kg]

HW2V = 663.375 + 1.63599*T2          ! Liquid
HW2L = 7.16036 + 2.63768*T2          ! Vapour

! --------------------------------------------------------------------------------------------------------------------

! Calculation of temperature and vapour holdup in LP evaporator

! Liquid and vapour enthalpies [J]

HL1 = WL1 * HW1L
```

```
HV1 = WV1 * HW1V

! Temperature in LP evaporator, using implicit solver

RESIDT1 = H1 - HL1 -HV1
T1 = IMPLC(RESIDT1,T1ic)

! Vapour holdup [kg], using implicit solver

RESIDWV1 = P1 - Z1*R*T1/(MW*VG1*100)
WV1 = IMPLC(RESIDWV1,WV1ic)

! ------------------------------------------------------------------------------------------------------------------------

! Calculation of temperature and vapour mass content in IP evaporator

! Liquid and vapour enthalpies [J]

HL2 = WL2 * HW2L
HV2 = WV2 * HW2V

! Temperature in IP evaporator, using implicit solver

RESIDT2 = H2 - HL2 -HV2
T2 = IMPLC(RESIDT2,T2ic)

! Vapour holdup [kg], using implicit solver

RESIDWV2 = P2 - Z2*R*T2/(MW*VG2*100)
WV2 = IMPLC(RESIDWV2,WV2ic)

! ------------------------------------------------------------------------------------------------------------------------

! Liquid holdup [kg]

WL1 = W1 - WV1              ! LP evaporator
WL2 = W2 - WV2              ! IP evaporator
W3 = W - W1 - W2 - W4       ! Receiver

! ------------------------------------------------------------------------------------------------------------------------
```

```
! Liquid refrigerant specific volume [m3/kg]

VF1 = 8.05418E-4 + 3.934E-6*T1
VF2 = 5.91479E-4 + 4.777E-6*T2
VF3 = -2.00589E-4 + 7.493E-6*TD

! Vapour volume [m3] in the vessels

V1G = V1 - WL1 * VF1                ! LP evaporator
V2G = V2 - WL2 * VF2                ! IP evaporator
V4G = V3 + VC - W3 * VF3            ! Condenser and receiver

! Vapour refrigerant specific volume [m3/kg]

VG1 = V1G/WV1
VG2 = V2G/WV2
VG4 = V4G/W4

! -------------------------------------------------------------------------------------------------------------------

! Refrigerant temperature in the condenser [K], using implicit solver

RESIDTD = P3 - Z3*R*TD/(MW*VG4*100)
TD = IMPLC(RESIDTD,TDic)

! -------------------------------------------------------------------------------------------------------------------

! Inter-stage mixing node pressure [bar], using implicit solver

RESIDPA = FG3 - FG1 - FG2
PA = IMPLC(RESIDPA,PAic)

! -------------------------------------------------------------------------------------------------------------------

! Compressibility factors

Z1 = 0.955 + 0.0*T1
Z2 = 0.566 + 1.26E-3*TB
Z3 = -0.194 + 2.97E-3*TD
```

```
! ---------------------------------------------------------------------------------------------------------------------------

! Volumetric holdups in the vessels [m3]
! Note that these values are used as indicators for levels in the vessels
! and hence are referred to as levels in the control scheme application

L1 = WL1*VF1                    ! LP evaporator
L2 = WL2*VF2                    ! IP evaporator
L3 = W3*VF3                     ! Receiver

! ---------------------------------------------------------------------------------------------------------------------------

! CONTROL IMPLEMENTATION
! ========================

! Note that L3 is not controlled in this scheme

! Error evaluation (ysp - y)

ER1 = L1SP - L1
ER2 = L2SP - L2
ER3 = P1SP - P1
ER4 = P2SP - P2
ER5 = P3SP - P3
ER6 = L3SP - L3
ERTP1 = TP1OSP - TP1O
ERTP2 = TP2OSP - TP2O

! Error integration

IER1 = INTEG(ER1,IER1ic)
IER2 = INTEG(ER2,IER2ic)
IER3 = INTEG(ER3,IER3ic)
IER4 = INTEG(ER4,IER4ic)
IER5 = INTEG(ER5,IER5ic)
IERTP1 = INTEG(ERTP1,IERTP1ic)
IERTP2 = INTEG(ERTP2,IERTP2ic)

! ISE evaluation
```

```
SE1=ER1*ER1
SE2=ER2*ER2
SE3=ER3*ER3
SE4=ER4*ER4
SE5=ER5*ER5
SE6=ER6*ER6
SETP1=ERTP1*ERTP1
SETP2=ERTP2*ERTP2

ISE1=INTEG(SE1,0)
ISE2=INTEG(SE2,0)
ISE3=INTEG(SE3,0)
ISE4=INTEG(SE4,0)
ISE5=INTEG(SE5,0)
ISE6=INTEG(SE6,0)
ISETP1=INTEG(SETP1,0)
ISETP2=INTEG(SETP2,0)

! Dummy "lin" variables are defined as they are needed in the linearisation of the model

CONSTANT XV2lin = 0
CONSTANT XV3lin = 0
CONSTANT Nlin = 0
CONSTANT XV1lin = 0
CONSTANT FCP3lin = 0

! Controllers actions

XV2  = K1 * (ER1 + IER1 / TI1) + XV2ic + XV2lin          ! L1-XV2 loop
XV3  = K2 * (ER2 + IER2 / TI2) + XV3ic + XV3lin          ! L2-XV3 loop
N    = K3 * (ER3 + IER3 / TI3) + Nic + Nlin              ! P1-N slave loop
XV1  = K4 * (ER4 + IER4 / TI4) + XV1ic + XV1lin          ! P2-XV2 loop
FCP3 = K5 * (ER5 + IER5 / TI5) + FCP3ic  + FCP3lin       ! P3-FCP3 loop

! Cascade temperature controller action, setting P1 set point
P1SP=0.9 + KT * (ERTP1 + IERTP1 / TIT)

! -----------------------------------------------------------------------------------------------------------------------
```

```
! Power consumption [J/s]

! Power calculated based on energy balance on each compressor

PTH1 = 0.5 * FG1* (CPO1+CPI1) * (TA-T1) * 1000/ MW
PTH2 = 0.5 * FG3* (CPO2+CPI2) * (TC-TB) * 1000/ MW
PTHTOTAL = PTH1 + PTH2

POWERINT = INTEG(PTHTOTAL,(1.4846e6))

! ------------------------------------------------------------------------------------------------------------------------

! Program termination
! Run-time [s]

CONSTANT TSTP = 3499.99
TERMT (T.GE.TSTP)

END             ! OF DERIVATIVE


! ==========================================================================
! ==========================================================================

END             ! OF DYNAMIC


! ==========================================================================
! ==========================================================================

END             ! OF PROGRAM
```